Práctica 8

Dinámica de las Caídas

8.1. Caída vertical en un medio resistente

Consideramos una partícula P de masa m sumergida en un fluido resistente al movimiento. Las fuerzas aplicadas sobre dicha partícula son: la fuerza de la gravedad $\mathbf{F}_g$, la fuerza de resistencia $\mathbf{F}_r$ que opone el medio al movimiento y el empuje $\mathbf{E}$ ejercido por el fluido desalojado por P. Si denotamos por $\mathbf{j}$ al vector unitario de sentido vertical hacia arriba, por m' la cantidad de masa del fluido desalojada por P y por v la velocidad de P; entonces, $\mathbf{F}_g = -mg\,\mathbf{j}$, $\mathbf{E} = m'g\,\mathbf{j}$ y $\mathbf{F}_r$ es opuesto a v y proporcional a v si el régimen del movimiento se supone *laminar*, es decir suave y sin remolinos, o $\mathbf{F}_r$ es proporcional a v^2 si el régimen de la caída se supone *turbulento*. En una caída vertical, todas las fuerzas intervinientes sobre P son verticales y su balance está dado por la igualdad vectorial:

$$m\frac{dv}{dt}\,\mathbf{j} = \mathbf{F} = \mathbf{F}_g + \mathbf{E} + \mathbf{F}_r = (-mg + m'g - kv|v|^{r-1})\,\mathbf{j} = (-g(m-m') - kv|v|^{r-1})\,\mathbf{j} \quad (8.1)$$

donde la potencia r es igual a 1 si el régimen del movimiento es *laminar*, e igual a 2 si el régimen es *turbulento* y donde la constante k es denominada *factor de proporcionalidad de la resistencia.*

De aquí se deduce la ecuación general de las caídas verticales en un medio resistente

$$\frac{dv}{dt} = -g(1-\frac{m'}{m}) - \frac{k}{m}\,v|v|^{r-1} = -g(1-\frac{\rho'}{\rho}) - \frac{k}{m}\,v|v|^{r-1} \quad (8.2)$$

donde ρ', ρ denotan las densidades del fluido y de P respectivamente y donde suponemos $\rho' < \rho$.

Para $a = g(1-\rho'/\rho)$, $b = k/m$ y $v_t = -(a/b)^{1/r} = -\left(mg/k\right)^{1/r}\left(1-\rho'/\rho\right)^{1/r}$, la ecuación diferencial anterior se expresa $dv/dt = -a - bv|v|^{r-1}$ o equivalentemente:

$$\frac{dv}{-a - bv|v|^{r-1}} = dt \Leftrightarrow \frac{dv}{1 + \frac{b}{a}\,v|v|^{r-1}} = -a\,dt \quad (8.3)$$

deduciéndose las siguientes expresiones, c y d constantes, para la velocidad v y la altura h del movimiento de la partícula:

Caída laminar:

$$\begin{array}{lcl} v' + bv = -a & \Leftrightarrow & v_G(t) = -a/b + c\,e^{-t\,b} = v_t + c\,e^{-t\,k/m} \\ h_G(t) = d + \int v_G(t)\,dt & \Leftrightarrow & h_G(t) = d + v_t\,t - \dfrac{c\,m}{k}\,e^{-t\,k/m} \end{array} \tag{8.4}$$

Subida turbulenta:

$$\frac{dv}{1+(v\,\sqrt{b/a})^2} = -a\,dt \;\Leftrightarrow \left\{ \begin{array}{l} v_G(t) = \sqrt{a/b}\ \mathrm{tg}(c - t\,\sqrt{ab}) = v_t\ \mathrm{tg}(c - t\,v_t k/m) \\ h_G(t) = d + \dfrac{m}{k}\ \ln|\cos(c - t\,v_t k/m)| \end{array} \right. \tag{8.5}$$

Caída turbulenta:

$$\frac{dv}{1-(v\,\sqrt{b/a})^2} = -a\,dt \;\overset{*}{\Leftrightarrow} \begin{array}{l} \left\{ \begin{array}{l} v_G(t) = \sqrt{a/b}\ \mathrm{th}(c - t\,\sqrt{ab}) = v_t\ \mathrm{th}(c - t\,v_t k/m) \\ h_G(t) = d - \dfrac{m}{k}\ \ln|\,\mathrm{ch}(c - t\,v_t k/m)| \end{array} \right. \\ \quad \text{ó} \\ \left\{ \begin{array}{l} v_G(t) = \sqrt{a/b}\ \mathrm{cth}(c - t\,\sqrt{ab}) = v_t\ \mathrm{cth}(c - t\,v_t k/m) \\ h_G(t) = d - \dfrac{m}{k}\ \ln|\,\mathrm{sh}(c - t\,v_t k/m)| \end{array} \right. \end{array} \tag{8.6}$$

donde

$$v_t = \lim_{t\to+\infty} v_G(t) = -\left(\frac{mg}{k}\right)^{1/r}\left(1 - \frac{\rho'}{\rho}\right)^{1/r}$$

recibe el nombre de *velocidad terminal* de la partícula P y donde th, cth son las funciones tangente y cotangente hiperbólicas.

Es importante observar que $v_G = v_t$ precisamente cuando el balance de fuerzas actuantes es cero, en cuyo caso y como es lógico, la partícula describe un movimiento rectilíneo uniforme. Desde el punto de vista matemático, enunciaríamos lo anterior diciendo que v_t es la única constante que es solución de la ecuación diferencial de las caídas.

Haciendo $u = v\,\sqrt{b/a}$ justificamos la equivalencia $*$ ya que:

$$-\sqrt{ab}\,dt = \frac{du}{1-u^2} \Leftrightarrow c - t\,\sqrt{ab} = \int \frac{du}{1-u^2} = \left\{ \begin{array}{ll} \mathrm{arcth}(u) & \text{si } |u| < 1 \\ \mathrm{arccth}(u) & \text{si } |u| > 1 \end{array} \right.$$

y de aquí obtenemos la fórmula de v_G dada en (8.6).

8.2. Elementos necesarios

En la primera línea cargamos los elementos de dibujo, la segunda instrucción nos permitirá dibujar campos:

```
> with(plots):   with(DEtools):
> macro(kk=(thickness=3,color=black,labels=["Tiempo, t",""]));
```

$$kk$$

8.3. Caída de una gota de lluvia

Práctica A Una gota de agua situada en una nube a 3000 ⟦m⟧ de altura comienza a caer. Sabiendo que su masa es 0.05 ⟦g⟧ y que el factor de resistencia con el aire es $5.5 \cdot 10^{-5}$ ⟦kg/s⟧, se pide:

1. Determinar la ecuación del movimiento de la gota de agua.
2. Determinar la velocidad terminal del movimiento.
3. Obtener la representación gráfica de la velocidad v y de la altura h en función del tiempo.
4. Calcular el tiempo que tarda en llegar al suelo.
5. Comprobar que para $t > 5$ el movimiento de la gota de agua es, en la práctica, indistinguible de un movimiento rectilíneo uniforme con velocidad v_t.

Al ser una partícula de pequeña masa supondremos que el régimen del movimiento es laminar y por tanto que

$$v' = -g - \frac{5.5 \cdot 10^{-5}}{5 \cdot 10^{-5}}\, v = -g - 1.1\, v; \qquad v(0) = 0$$

es la ecuación diferencial del movimiento de nuestra gota de lluvia ($\rho'/\rho < 0.002 \simeq 0$), $v_t = -\dfrac{mg}{k} = -8.91$ ⟦m/s⟧ su velocidad terminal, $v(t) = -8.91(1 - e^{-1.1\,t})$ la función velocidad, $h(t) = 3000 + \int_0^t v(s)\, ds = 3008.10 - 8.91\, t - 8.10 \cdot e^{-1.1\,t}$ la función altura y $t_s = 337.6438$ ⟦s⟧ el tiempo que tarda en llegar al suelo. Finalmente, el último punto, resulta de que el error relativo entre $v(t) = v_t(1 - e^{-1.1\,t})$, $t > 5$, y v_t vale

$$\varepsilon_r = \frac{|v_t - v(t)|}{|v_t|} = e^{-1.1\,t} < e^{-5.5} < 0.005 = 0.5\,\% \tag{8.7}$$

y de ello, se deduce que el movimiento de la gota es indistinguible, a partir del quinto segundo, de un movimiento rectilíneo uniforme de velocidad v_t, lo que es fácilmente observable a partir de la gráfica de la función v, figura 8.1.

La solución a todos los apartados lo podemos hacer con el listado:

```
> g:= 9.8;
```

$$g := 9.8$$

```
> m:= 5e-5; k:= 5.5e-5;
```

$$m := 0.00005$$
$$k := 0.000055$$

```
> vt:= -g*m/k;
> vt*3.6*km/h;
```

$$vt := -8.909090909$$
$$-\frac{32.07272727\, km}{h}$$

```
> eq:= diff(v(t),t)= -g-k/m*v(t);
> Sol:=dsolve({eq,v(0)=0},{v(t)});
> vg:=subs(Sol,v(t));
> evalf(%);
```

$$eq := \frac{d}{dt}\,\mathrm{v}(t) = -9.8 - 1.100000000\,\mathrm{v}(t)$$
$$Sol := \mathrm{v}(t) = -\frac{98}{11} + \frac{98}{11}\,e^{(-\frac{11\,t}{10})}$$
$$vg := -\frac{98}{11} + \frac{98}{11}\,e^{(-\frac{11\,t}{10})}$$
$$-8.909090909 + 8.909090909\,e^{(-1.100000000\,t)}$$

Por vg denotamos la función velocidad de la gota de y por hg la función altura

```
> hg:= 3000+int(vg,t=0..t);
> evalf(%);
```

$$hg := \frac{363980}{121} - \frac{98\,t}{11} - \frac{980}{121}\,e^{(-\frac{11\,t}{10})}$$
$$3008.099174 - 8.909090909\,t - 8.099173554\,e^{(-1.100000000\,t)}$$

```
> ts:= fsolve(hg,t=100);
```

$$ts := 337.6437848$$

```
> plot(vg,t=0..10,kk);
```



Figura 8.1: Velocidad de caída de una gota de lluvia.

```
> plot(hg,t=0..10,kk);
```



Figura 8.2: Función altura para la caída de una gota de lluvia.

8.3.1. Interpretación geométrica de solución de una ecuación diferencial

Dada una función $v(t)$, $t \in [0,10]$, el vector tangente en (t,v) a la curva

$$\begin{array}{rccc} \Gamma_v & : & [0,10] & \longrightarrow & \mathbb{R}^2 \\ & & t & \mapsto & (t, v(t)) \end{array}$$

es $(1, v')$. En particular, $v(t)$, $t \in [0,10]$, es la solución a la ecuación diferencial $v' = -g - 1.1v$, $v(0) = v_0$, precisamente si Γ_v es lo que se denomina la única *curva integral* del campo $F = \frac{\partial}{\partial t} - (g + 1.1v)\frac{\partial}{\partial v}$ que pasa por el punto $(0, v_0)$, es decir, si Γ_v es la única curva que pasa por $(0, v_0)$ y cuyo vector tangente en cada punto (t, v) es el vector $(1, -g - 1.1v)$.

Obtener la representación gráfica del campo F y de las curvas integrales que pasan por $(0,0)$, $(1,3)$, $(0,-15)$, $(0,-20)$ y $(5,5)$ es bien sencillo, basta con escribir las instrucciones:

```
> Feq:= [diff(T(t),t)=1, eq];  #Campo de la Ecuación eq
```

$$Feq := [\frac{d}{dt}\,\mathrm{T}(t) = 1,\ \frac{d}{dt}\,\mathrm{v}(t) = -9.8 - 1.100000000\,\mathrm{v}(t)]$$

```
> DEplot(Feq,[T(t),v(t)],
> t=0..40, T=0..20, v=-20..5,
> [[T(0)=0,v(0)=0],[T(0)=1,v(0)=3],[T(0)=0,v(0)=-15],
> [T(0)=4,v(0)=-20],[T(0)=5,v(0)=5]]);
```



Figura 8.3: Espacio de flujo de la ecuación $v' = -g - 1.1v$.

Procediendo de esta forma podemos realizar un estudio cualitativo de las soluciones de la ecuación diferencial comprobando cuál es el distinto comportamiento que éstas tienen ante diferentes condiciones iniciales (t_0, v_{t_0}). El plano (T, V) en el que hemos realizado la representación se denomina *espacio de fases*. Si además, como en nuestro caso aparece la representación del campo F; entonces, de la figura obtenida se dice que es el *espacio de flujo* de la ecuación diferencial $v' = -g - 1.1v$.

8.4. Caída de un paracaidista

Práctica B Un paracaidista se deja caer desde un helicóptero situado a 4000 $[\![\mathrm{m}]\!]$. Disfruta de las emociones que la caída libre produce durante 60 $[\![\mathrm{s}]\!]$ y después acciona la apertura de su paracaídas. Sabiendo que su masa es de 60 $[\![\text{kg}]\!]$, que su factor de resistencia con el aire es de 0.2 $[\![\text{kg/m}]\!]$ y de 7.8 $[\![\text{kg/m}]\!]$ cuando abre el paracaídas, se pide:

1. Calcular la velocidad terminal del movimiento con y sin paracaídas.
2. Determinar la ecuación del movimiento del paracaidista con y sin paracaídas.
3. Calcular la altura a la que se encuentra cuando abre el paracaídas.
4. Obtener la representación gráfica de la velocidad, aceleración y altura en los 200 primeros segundos.
5. Calcular el tiempo total que tarda en llegar al suelo.
6. Realizar una representación gráfica del espacio de flujo asociado a la ecuación del movimiento con paracaídas.

El régimen de caída de un paracaidista se supone turbulento. En consecuencia, las soluciones a los anteriores apartados son:

1) La velocidad terminal sin paracaídas es $v_t = -\sqrt{mg/k} = -\sqrt{9.8 \cdot 60/0.2} = -54.22\,[\![\text{m/s}]\!] = -195.20\,[\![\text{km/h}]\!]$. Con paracaídas $V_t = -\sqrt{9.8 \cdot 60/1.2} = -8.68\,[\![\text{m/s}]\!] = -31.26\,[\![\text{km/h}]\!]$.

2) Si denotamos por $V(t)$ y $v(t)$ las velocidades con y sin paracaídas; entonces, las ecuaciones del movimiento son:

$$\begin{array}{llll} v' = -g + \frac{0.2}{60}v^2 & = -g + \frac{1}{300}v^2, & v(0) = 0, & \text{Sin Paracaídas} \\ V' = -g + \frac{7.8}{60}V^2 & = -g + \frac{13}{100}V^2, & V(60) = -54.22, & \text{Con Paracaídas} \end{array} \tag{8.8}$$

3) y 4) $v(t) = -54.22\,\text{th}(t\,54.22 \cdot 0.2/60) = -54.22\,\text{th}(0.18\,t)$, $t \in [0, 60]$ y $V(t) = -8.68\,\text{cth}(c - t\,V_t \cdot 7.8/60) = -8.68\,\text{cth}(c + 1.12\,t)$, $t \geq 60$, donde c hay que determinarla por la condición $V(60) = -8.68\,\text{cth}(c - V_t\,7.8) = -54.22$ resultando $c = \text{arcth}(V_t/v_t) + 7.8\,V_t = -67.56$ y, por tanto:

$$\begin{array}{lll} v(t) & = -54.22\ \text{th}(0.18\,t), & t \in [0, 60] \\ V(t) & = -\ 8.68\ \text{cth}(-67.56 + 1.12\,t), & t \geq 60 \end{array}$$

y también:

$$\begin{array}{ll} ac(t) = -g + \frac{1}{300}v(t)^2 = -g(1 - \text{th}(0.18\,t)), & t \in [0, 60] \\ Ac(t) = -g + \frac{13}{100}V(t)^2 = -g(1 - \text{cth}(-67.56 + 1.12\,t)) & t \geq 60 \end{array}$$

para $Ac(t)$, $ac(t)$, la función aceleración del paracaidista con y sin paracaídas.

Así mismo, si denotamos por $z(t)$, $Z(t)$ las funciones altura para $t \in [0, 60]$, $t \geq 60$, respectivamente; entonces:

$$\begin{array}{llll} z(t) & = 4000 + \int_0^t v(s)\,ds & = 4000 - 300\ln(\text{ch}(0.18\,t)), & t \in [0, 60] \\ z(60) & = 954.64\,[\![\text{m}]\!] & \text{altura de apertura del paracaídas} & \\ Z(t) & = h(60) + \int_{60}^t V(s)\,ds & = 940.65 - \frac{100}{13}\ln(\text{sh}\,|-67.56 + 1.12\,t|), & t \geq 60 \end{array}$$

5) Se trata de resolver la ecuación $940.65 = \frac{100}{13}\ln(\text{sh}(-67.56 + 1.12\,t))$ de donde se obtiene $t_s = (\text{arcsh}(e^{13 \cdot 9.4065}) + 67.56)/1.12 = 168.81\,[\![\text{s}]\!]$.

Finalmente, la gráfica del punto 6 es la figura 8.7. Las cuentas de todos los apartados las podemos encontrar en el listado que viene a continuación. En ellas podemos comprobar que hemos preferido resolver numéricamente la ecuación diferencial debido a la dificultad en la elección de la determinación correcta, pues según sabemos por las ecuaciones (8.5) y (8.6), debemos elegir entre tres posibilidades:

```
> g:=9.8;
```

$$g := 9.8$$

```
> m:= 60;
```

$$m := 60$$

```
> tA:= 60; # tA tiempo de apertura
```

$$tA := 60$$

```
> ks:= 0.2;  kc:=7.8;
> #  factores de resistencia SIN
> #  y CON paracaídas
```

$$ks := 0.2$$

$$kc := 7.8$$

```
> vt:= -sqrt(g*m/ks);%*3.6*km/h; # Sin Paracaidas
> Vt:= -sqrt(g*m/kc);%*3.6*km/h; # Con Paracaidas
```

$$vt := -54.22176685$$

$$-\frac{195.1983607\,km}{h}$$

$$Vt := -8.682431421$$

$$-\frac{31.25675312\,km}{h}$$

```
> eqS:= diff(z(t),t$2) =-g-ks/m*diff(z(t),t)*abs(diff(z(t),t));
> # Sin Paracaidas
> eqC:= diff(z(t),t$2) =-g-kc/m*diff(z(t),t)*abs(diff(z(t),t));
> # Con Paracaidas
```

$$eqS := \frac{d^2}{dt^2}\,\mathrm{z}(t) = -9.8 - 0.003333333333\,(\tfrac{d}{dt}\,\mathrm{z}(t))\,\left|\tfrac{d}{dt}\,\mathrm{z}(t)\right|$$

$$eqC := \frac{d^2}{dt^2}\,\mathrm{z}(t) = -9.8 - 0.1300000000\,(\tfrac{d}{dt}\,\mathrm{z}(t))\,\left|\tfrac{d}{dt}\,\mathrm{z}(t)\right|$$

```
> Sol:= dsolve({eqS, z(0)=4000, D(z)(0)=0
> },z(t),type=numeric,output=listprocedure);
> zs_:= subs(Sol,z(t));
> vs_:= subs(Sol,diff(z(t),t)); evalf(%);
> # vs_ función velocidad SIN paracaidas
> # zs_ función altura SIN paracaidas
```

$$Sol := [t = (\mathbf{proc}(t)\ \ldots\ \mathbf{end\ proc}\ ),\ \mathrm{z}(t) = (\mathbf{proc}(t)\ \ldots\ \mathbf{end\ proc}),$$
$$\tfrac{d}{dt}\,\mathrm{z}(t) = (\mathbf{proc}(t)\ \ldots\ \mathbf{end\ proc})]$$

$$zs\_ := \mathbf{proc}(t)\ \ldots\ \mathbf{end\ proc}$$

$$vs\_ := \mathbf{proc}(t)\ \ldots\ \mathbf{end\ proc}$$

$$\mathbf{proc}(t)\ \ldots\ \mathbf{end\ proc}$$

```
> h1:= zs_(60);
```

$$h1 := 954.638136966303592$$

h1=954.64[m] es la altura a la que se abre el paracaidas

```
> Sol:= dsolve({eqC, z(tA)=h1, D(z)(tA)=vt
> },z(t),type=numeric,output=listprocedure);
> zc_:= subs(Sol,z(t));
> vc_:= subs(Sol,diff(z(t),t)); evalf(%);
> # vc_ función velocidad CON paracaidas
> # zs_ función altura CON paracaidas
```

$$Sol := [t = (\mathbf{proc}(t)\ \ldots\ \mathbf{end\ proc}\ ),\ \mathrm{z}(t) = (\mathbf{proc}(t)\ \ldots\ \mathbf{end\ proc}),$$
$$\frac{d}{dt}\,\mathrm{z}(t) = (\mathbf{proc}(t)\ \ldots\ \mathbf{end\ proc})]$$

$$zc_- := \mathbf{proc}(t)\ \ldots\ \mathbf{end\ proc}$$

$$vc_- := \mathbf{proc}(t)\ \ldots\ \mathbf{end\ proc}$$

$$\mathbf{proc}(t)\ \ldots\ \mathbf{end\ proc}$$

```
> as_:= -g+ ks/m* vs_^2: evalf(%); #   Aceleraciones CON y SIN
> ac_:= -g+ kc/m* vc_^2: evalf(%); #   paracaídas
```

$$-9.8 + 0.003333333333\, vs_-^2$$

$$-9.8 + 0.1300000000\, vc_-^2$$

```
> A:=plot(vs_(t),t=0..tA,kk): B:= plot(vc_(t),t=tA..tA+20,kk,
> color=blue):
> display(A,B);
```



Figura 8.4: Velocidad de caída de un paracaidista.

```
> A:= plot(as_(t),t=0..tA,kk): B:=
> plot(ac_(t),t=tA+0.3..tA+20,kk,color=blue):
> display(A,B);
```



Figura 8.5: Aceleración en la caída de un paracaidista.

```
> A:= plot(zs_(t),t=0..tA,kk): B:=
> plot(zc_(t),t=tA..200,kk,color=blue):
> display(A,B);
```



Figura 8.6: Función altura para la caída de un paracaidista.

```
> ts:= fsolve(zc_(t),t=160);
```

$$ts := 168.8101736$$

ts=168.81[s] tiempo de caída del paracaidista

```
> eqC:= diff(v(t),t)=-g+kc/m*v(t)^2;
> FeqC:= [diff(T(t),t)=1,eqC];
```

$$eqC := \frac{d}{dt}\,\mathrm{v}(t) = -9.8 + 0.1300000000\,\mathrm{v}(t)^2$$

$$FeqC := [\frac{d}{dt}\,\mathrm{T}(t) = 1,\ \frac{d}{dt}\,\mathrm{v}(t) = -9.8 + 0.1300000000\,\mathrm{v}(t)^2]$$

```
> DEplot(FeqC,[T(t),v(t)],
> t=0..30, T=0..30, v=-30..5,
> [[T(0)=0,v(0)=0],[T(0)=7,v(0)=5],[T(0)=5,v(0)=-30],
> [T(0)=0,v(0)=-30],[T(0)=15,v(0)=-30]]);
```



Figura 8.7: Espacio de flujo de la ecuación diferencial $v' = -g - 0.13v|v|$.

8.5. Caída semiparabólica en un medio resistente

Práctica C Un paracaidista se lanza de un helicóptero situado a 1000 ⟦m⟧ con una velocidad horizontal de 5 ⟦m/s⟧ a la vez que acciona la apertura de su paracaídas. Sabiendo que su masa es de 60 ⟦kg⟧ y que su factor de resistencia con el aire es 7.8 ⟦kg/m⟧, se pide:

1. Calcular y representar la velocidad del movimiento del paracaidista.
2. Representar la trayectoria del paracaidista.
3. Calcular el tiempo total que tarda en llegar al suelo.
4. Calcular el desplazamiento horizontal de la posición de caída respecto a la vertical con el helicóptero.
5. Realizar una representación gráfica del espacio de flujo asociado a la ecuación diferencial del movimiento.

Para explicar el movimiento necesitamos dos componentes, la horizontal x, con vector unitario $\mathbf{i}$, y la vertical z, con vector unitario $\mathbf{j}$. Ahora, la expresión de la fuerza de resistencia del aire toma la forma:

$$\mathbf{F}_r = -7.8|v|v = -7.8\sqrt{v_x^2 + v_z^2}\,(v_x\mathbf{i} + v_z\mathbf{j}) \tag{8.9}$$

con lo que el balance de fuerzas se expresa:

$$m(\frac{dv_x}{dt}\,\mathbf{i} + \frac{dv_z}{dt}\,\mathbf{j}) = -7.8\sqrt{v_x^2 + v_z^2}\,v_x\mathbf{i} + (-g - 7.8\sqrt{v_x^2 + v_z^2}\,v_z)\,\mathbf{j} \tag{8.10}$$

deduciéndose que la ecuación del movimiento se escribe:

$$\left\{\begin{array}{rcl} \dfrac{dv_z}{dt} & = & -g - 0.13\sqrt{v_x^2 + v_z^2}\,v_z, \quad v_z(0) = 0, \\ \dfrac{dv_x}{dt} & = & -0.13\sqrt{v_x^2 + v_z^2}\,v_x, \quad v_x(0) = 5, \end{array}\right. \tag{8.11}$$

y también:

$$\begin{cases} d^2z/dt^2 = & -g - 0.13\sqrt{(dx/dt)^2 + (dz/dt)^2}\, dz/dt, \quad v_z(0) = 0, \quad z(0) = 1000, \\ d^2x/dt^2 = & -0.13\sqrt{(dx/dt)^2 + (dz/dt)^2}\, dx/dt, \quad v_x(0) = 5, \quad x(0) = 0 \end{cases} \tag{8.12}$$

Este sistema de ecuaciones diferenciales sólo puede ser resuelto utilizando métodos numéricos. Para resolverlo y para obtener la solución a todos los apartados, incluidas las representaciones 8.8, 8.9 y 8.10, realizamos el listado:

```
> vz:=diff(z(t),t): vx:= diff(x(t),t):
> ec1:= diff(z(t),t$2)=-g-0.13*sqrt(vz^2+vx^2)*vz;
> ec2:= diff(x(t),t$2)=  -0.13*sqrt(vz^2+vx^2)*vx;
```

$$ec1 := \frac{d^2}{dt^2}\,\mathrm{z}(t) = -9.8 - 0.13\,\sqrt{(\tfrac{d}{dt}\,\mathrm{z}(t))^2 + (\tfrac{d}{dt}\,\mathrm{x}(t))^2}\,(\tfrac{d}{dt}\,\mathrm{z}(t))$$

$$ec2 := \frac{d^2}{dt^2}\,\mathrm{x}(t) = -0.13\,\sqrt{(\tfrac{d}{dt}\,\mathrm{z}(t))^2 + (\tfrac{d}{dt}\,\mathrm{x}(t))^2}\,(\tfrac{d}{dt}\,\mathrm{x}(t))$$

```
> Ini:= z(0)=1000, x(0)=0,D(z)(0)=0, D(x)(0)=5;
```

$$Ini := \mathrm{z}(0) = 1000,\ \mathrm{x}(0) = 0,\ \mathrm{D}(z)(0) = 0,\ \mathrm{D}(x)(0) = 5$$

```
> Sol:=dsolve({ec1,ec2,Ini},{z(t),x(t)},
> type=numeric, output=listprocedure);
```

$$Sol := [t = (\mathbf{proc}(t)\ \ldots\ \mathbf{end\ proc}\ ),\ \mathrm{x}(t) = (\mathbf{proc}(t)\ \ldots\ \mathbf{end\ proc}),$$
$$\tfrac{d}{dt}\,\mathrm{x}(t) = (\mathbf{proc}(t)\ \ldots\ \mathbf{end\ proc}),\ \mathrm{z}(t) = (\mathbf{proc}(t)\ \ldots\ \mathbf{end\ proc}),$$
$$\tfrac{d}{dt}\,\mathrm{z}(t) = (\mathbf{proc}(t)\ \ldots\ \mathbf{end\ proc})]$$

```
> z_:=subs(Sol,z(t)); x_:=subs(Sol,x(t));
> vz_:=subs(Sol,diff(z(t),t));
> vx_:=subs(Sol,diff(x(t),t));
> vz_(0);  vx_(0); vx_(10);
```

$$z\_ := \mathbf{proc}(t)\ \ldots\ \mathbf{end\ proc}$$
$$x\_ := \mathbf{proc}(t)\ \ldots\ \mathbf{end\ proc}$$
$$vz\_ := \mathbf{proc}(t)\ \ldots\ \mathbf{end\ proc}$$
$$vx\_ := \mathbf{proc}(t)\ \ldots\ \mathbf{end\ proc}$$
$$0.$$
$$5.$$
$$0.0001030159243350234 28$$

Con las tres siguientes instrucciones obtenemos las soluciones a 3) y 4): el tiempo de caída es $\boxed{t_s = 115.84\,[\![\mathrm{s}]\!]}$, el desplazamiento horizontal total es $\boxed{5.90\,[\![\mathrm{m}]\!]}$ y la velocidad total final es $\boxed{8.68\,[\![\mathrm{m/s}]\!]}$.

```
> ts:= fsolve(z_(t),t=0..200); # Tiempo de caída
```

$$ts := 115.8412852$$

```
> x_(ts);  #  Cálculo del desplazamiento horizontal sobre
> #  la vertical del helicoptero
```

$$5.90343360366109149$$

```
> VelocidadTotalSuelo:= sqrt(vx_(ts)^2+vz_(ts)^2);
> vz_(ts);   # Componente vertical de v en ts
> Vt;        # Velocidad terminal anterior
```

$$VelocidadTotalSuelo := 8.682432309$$

$$-8.68243230884917060$$

$$-8.682431421$$

```
> plot([vx_,vz_],0..10,legend=["Vx","Vz"],kk,linestyle=[solid,
> dot]);
```



Figura 8.8: Componentes v_x, v_z de la velocidad del paracaidista.

```
> P_r:= [[x_(n*0.2),z_(n*0.2)] $n=0..50]:# Puntos de caída cada 0.2[s]
> X_:=t->5*t; Z_:=t->1000-0.5*g*t^2;
> P_s:= [[X_(n*0.2),Z_(n*0.2)] $n=0..8]; # Puntos de caída SIN
> #resistencia del aire
```

$$X\_ := t \to 5\,t$$

$$Z\_ := t \to 1000 - 0.5\,g\,t^2$$

P_s := [[0., 1000.], [1.0, 999.8040], [2.0, 999.2160], [3.0, 998.2360], [4.0, 996.8640], [5.0, 995.1000], [6.0, 992.9440], [7.0, 990.3960], [8.0, 987.4560]]

```
> A:=plot( P_r,style=point,symbol=circle,kk,legend="Con
> Resistencia",labels=["Desp. Horizontal","Altura"]):
> B:=plot( P_s, style=point,symbol=diamond,kk,legend="Sin
> Resistencia",labels=["Desp. Horizontal","Altura"]):
> display(A,B);
```



Figura 8.9: Posiciones del paracaidista cada 0.2 ⟦s⟧ con y sin resistencia del aire.

```
> Ts:= fsolve(Z_=0,0..100); # tiempo de caída SIN resistencia
> # del aire
```

$$Ts := 14.28571429$$

```
> X_(Ts);    #  desplazamiento horizontal SIN
> #  resistencia del aire
```

$$71.42857145$$

```
> V:= sqrt(Vx(t)^2+Vz(t)^2);
> Feq:=[diff(Vx(t),t)=-kc/m*Vx(t)*V,diff(Vz(t),t)=-g-kc/m*Vz(t)*V];
```

$$V := \sqrt{\mathrm{Vx}(t)^2 + \mathrm{Vz}(t)^2}$$

$$Feq := [\tfrac{d}{dt}\,\mathrm{Vx}(t) = -0.1300000000\,\mathrm{Vx}(t)\,\sqrt{\mathrm{Vx}(t)^2 + \mathrm{Vz}(t)^2},$$
$$\tfrac{d}{dt}\,\mathrm{Vz}(t) = -9.8 - 0.1300000000\,\mathrm{Vz}(t)\,\sqrt{\mathrm{Vx}(t)^2 + \mathrm{Vz}(t)^2}]$$

```
> DEplot(Feq,[Vx(t),Vz(t)],
> t=0..40, Vx=0..5, Vz=-20..5,
> [[Vz(0)=0,Vx(0)=5],[Vz(0)=5,Vx(0)=5],
> [Vz(0)=-4,Vx(0)=5],[Vz(0)=-7,Vx(0)=5],
> [Vz(0)=-15,Vx(0)=5],[Vz(0)=-19,Vx(0)=5],
> [Vz(0)=5,Vx(0)=3] ]);
```



Figura 8.10: Flujo de la velocidad en la caída semiparabólica.

8.6. Ejercicios

Práctica D Supongamos en el enunciado de la práctica (8.A) que cuando cae la gota de lluvia la nube en la que se encuentra se ve sometido a un movimiento ascendente de 10 $[\![\mathrm{m/s}]\!]$ de velocidad constante. Calcular la máxima altura que llega a alcanzar la gota de agua.

Práctica E En una caída en régimen turbulento, el factor k de resistencia verifica

$$k = \frac{\rho' A\delta}{2} \tag{8.13}$$

para ρ' la densidad del fluido, A el valor del área de la mayor sección transversal de la partícula P en caída y δ un coeficiente que depende de la forma de P y que se denomina *coeficiente de arrastre*.

Para el caso de un paracaidista en situaciones normales, $\delta = 1.2$ y $\rho' = 1\,[\![\mathrm{kg/m^3}]\!]$. Teniendo esto en cuenta:

1. Calcular el radio R que debe tener el paracaídas de la práctica (8.B) suponiendo que es semiesférico.
2. Calcular el menor radio R que debe tener un paracaídas semiesférico si queremos lanzar un objeto de $100\,[\![\mathrm{kg}]\!]$ desde $3000\,[\![\mathrm{m}]\!]$ y que su velocidad terminal no sea superior a $20\,[\![\mathrm{km/h}]\!]$.

Práctica F Suponer en el enunciado de la práctica (8.C) que son dos los paracaidistas que caen al mismo tiempo y con las mismas condiciones, excepto que el segundo se deja caer sin ninguna velocidad inicial. ¿Llegarán al mismo tiempo al suelo?. Razona la respuesta.

Práctica G Calcular la máxima aceleración que soporta el paracaidista de la práctica (8.B). ¿Consideras que el valor obtenido es real?. Razona la respuesta.

Práctica H La tolerancia humana frente a elevadas aceleraciones es una preocupación común para médicos, agencias espaciales y militares, equipos de fórmula 1, etc. Realiza una pequeña investigación sobre el tema y con los resultados obtenidos realiza un documento (de texto, presentación,...) de carácter divulgativo pero con base científico-técnica.

Práctica I Un nuevo planeta ha sido descubierto en los límites de una galaxia lejana. Los exploradores tienen como primera misión el estudio de una extraña estructura con forma de 'T' que se eleva $500\,[\![\mathrm{m}]\!]$ por encima del único mar del planeta. El explorador encargado sufre la avería de su transporte en el extremo superior derecho de la estructura y, como consecuencia, queda aislado y sin la posibilidad de un fácil rescate en un momento crítico. Después de un corto cálculo por parte del comandante de la misión, éste decide que la mejor solución es persuadir a su compañero de que se deje caer.

Comprobar si es correcta la decisión del comandante sabiendo que la máxima sección del explorador es $A = 0.8\,[\![\mathrm{m}^2]\!]$, que su coeficiente de arrastre es $\delta = 0.9$, que su masa es $65\,[\![\mathrm{kg}]\!]$, que la gravedad del planeta es $5.2\,[\![\mathrm{m/s}^2]\!]$ y que la densidad de su aire es $\rho' = 4\,[\![\mathrm{kg/m}^3]\!]$.

Calcular también el tiempo total que tarda nuestro explorador en caer sobre la superficie del mar.

Práctica J Calcular el número de metros de anticipación que sobre la vertical debía tener en cuenta el artillero de un bombardero de la segunda guerra mundial para impactar sobre un objetivo con una bomba de $600\,[\![\mathrm{kg}]\!]$ teniendo en cuenta que el factor de resistencia de las bombas era de $3.6\,[\![\mathrm{kg/m}]\!]$, que el avión volaba a $500\,[\![\mathrm{km/h}]\!]$ y a una altura de $4000\,[\![\mathrm{m}]\!]$.

Práctica K Para $f(t,x)$ una función derivable en sus dos componentes, ¿puede ocurrir que la figura 8.11 sea la representación de dos soluciones de la ecuación diferencial $\dfrac{dx}{dt} = f(t,x)$ en su espacio de fases (T,X)?.

Práctica L Los datos dados en la práctica (8.A) son reales, a partir de ellos estimar la viscosidad del aire.



Figura 8.11: ¿Puede ser el espacio de fases de $x' = f(t,x)$?

8.7. Ejercicios de prácticas anteriores

Práctica M La tolerancia en el error para el presente ejercicio es de tres milésimas. De la intensidad de corriente $i(t)$ que circula por un circuito eléctrico se sabe que su medida en amperios viene dada, en función del tiempo t, por la función:

$$i(t) = 4.2 \cdot e^{-9.5\,t} + 1.34 \cdot \operatorname{sen}(50\,t - 1.11); \qquad t \text{ en segundos}$$

1. Realiza la representación gráfica de la función intensidad para $t \in [0,1]$.
2. Calcula la máxima intensidad que soporta el circuito, es decir, el máximo de la función $|i(t)|$.

Práctica N Consideramos la función $f\colon [0,1] \to \mathbb{R}$ definida por la igualdad

$$f(x) = \ln(1+5x) - 2x^2$$

Calcular, cometiendo un error menor a 5 milésimas, el valor $x = d$ para el cual $f(x)$ alcanza su único máximo en $[0,1]$.

Práctica Ñ Sea $x(t)$ la única solución de la ecuación diferencial

$$\left\{ \begin{array}{rcl} \dfrac{d^2x}{dt^2}(t) + 18x(t) & = & 80\cos(5t) \\ x(0) & = & 4 \\ \dfrac{dx}{dt}(0) & = & 0 \end{array} \right.$$

Para t en el intervalo $[0,10]$, se pide calcular, cometiendo un error menor a 5 milésimas, el valor máximo de la función $|x(t)|$, valor absoluto de $x(t)$.